


# Sheffield Economic Research Paper Series

**SERP Number: 2004002**



**Scott McDonald (University of Sheffield) and Karen Thierfelder (United States Naval Academy)**

**Deriving a Global Social Accounting Matrix from GTAP version 5 Data**

**March 2004**

Department of Economics  
University of Sheffield  
9 Mappin Street  
Sheffield  
S1 4DT  
United Kingdom  
[www.shef.ac.uk/economics](http://www.shef.ac.uk/economics)

## Abstract<sup>1</sup>

*This paper reports a method for converting version 5 of the GTAP database into a global SAM that is stored in GAMS data exchange (GDX) format or in a Header Array (HAR) format. The focus of attention is on the generation of a SAM representation of the GTAP database that is fully consistent with the GTAP model; as such the resultant SAM can be readily used to calibrate a version of the GTAP model that had been coded in GAMS. The GAMS code used to generate the SAM representation of the GTAP database is provided as a GAMS model library – this code has been tested with versions 5.0 and 5.3 of the GTAP database.*

**Keywords:** Social Accounting Matrix, Trade Analysis, Computable General Equilibrium.

Related materials for this paper are available via the following URL:

<http://www.shef.ac.uk/training/devectes/research/gatpsam.zip>

### Correspondence Address:

Dr Scott McDonald, Department of Economics, The University of Sheffield, 9 Mappin Street, Sheffield, S7 1HR, UK. tel: +44 114 2223407, email: s.mcdonald@sheffield.ac.uk

---

<sup>1</sup> The authors gratefully acknowledge the many explanations of the GTAP model and GEMPACK code provided by Terrie Walmsley; without these explanations this research would have been very much more difficult.

## Introduction

The GTAP database represents a major resource for economists who engage in multi-sector global modeling. The database is supplied in the Header Array (HAR) file format, which is used by GEMPACK, RunGTAP and the associated software developed for/by the GTAP project. However, while the HAR file format is very efficient it is not immediately usable by modeling programmes written using different software, e.g., GAMS, and although some modelers may wish to use the GTAP data, they may not wish to use the GTAP model and/or GEMPACK. One alternative is the GTAPnGAMS suite developed by Rutherford (1998), which provides both a means of transferring the data into a GAMS readable format and a representation of the GTAP model in GAMS. But while it is possible to use the programmes associated with GTAPnGAMS, in particular SEEHAR and HAR2GAMS,<sup>2</sup> to transfer GTAP data into a GAMS readable format there may be occasions when modelers wish to transform the GTAP database.

One potential transformation of the GTAP database is its conversion from an input-output based format with additional sub matrices for trade and other transactions into a Social Accounting Matrix (SAM) format. Since the databases for ALL whole economy models can always be represented in a SAM format, and some modelers choose to present their model databases in SAM format, such a transformation has attractions. The transformation of the database to a SAM format does however involve some complications. These arise, for the most part, from the fact that the GTAP database does not appear, at first sight, to observe the ‘law of one price’ in the rows of the input-output tables, i.e., the purchaser prices are not common across the rows of the input-output tables, due to the tax specification in GTAP.

The approach taken here to address the complications presented by the apparent absence of the ‘law of one price’ is to value entries in the commodity rows of the SAM at ‘sellers’ prices, thereby ensuring that all the entries in each commodity account row are valued at a common price. The purchasing agents are then assumed to make a separate payment to the government with respect to indirect taxes on commodities purchased for consumption/demand by domestic agents. This approach ensures that the national accounting identities for the SAM

---

<sup>2</sup> The SEEHAR software was written by Ken Pearson. Full details are available from <http://www.monash.edu.au/policy/gpf90.htm>. GAMS2HAR (and HAR2GAMS) were written by Mark Horridge and Tom Rutherford. Full details are available from <http://debreu.colorado.edu/gdxhar/index.html>.

are maintained. This process transforms the GTAP database into a global SAM format that is consistent with the structure of the price and tax systems found in the GTAP model.

Many SAM based CGE models are not set up to deal with domestic commodity demands that are valued at ‘sellers’ prices. Consequently the model requires altering or the database needs transforming if a SAM based CGE implementation is required. There are several ways in which the SAM could be transformed so that domestic demands are recorded in transactions prices so that the ‘law of one price’ is observed in the purchaser prices. All of these potential transformations rest upon the treatment of indirect taxes on commodities purchased for consumption/demand by domestic agents; for the benefit of users who do wish to produce a SAM where domestic demand transactions are recorded in purchaser prices the GAMS code is set up to generate estimates of the *ad valorem* indirect tax rates on commodities. A number of alternative SAM representations of the GTAP global SAM, and the associated GAMS code, are available in McDonald (2004), while some possible developments of the database that may enrich the representation of global economic transactions captured by the GTAP database are reported in McDonald and Sonmez (2004).

Another complication is introduced by differences in the terminology used in the GTAP literature and a common variant of that used in the SAM literature. Whereas the SAM literature refers to purchaser, basic, etc., prices, the GTAP literature refers to Agent, Market and World prices. For an imported commodity, the values of transactions at market prices are defined as the values at world prices plus trade and transport margins plus import duties, and the values at agents prices are then defined as the values at market prices plus any sales/purchase taxes. For domestic commodities, the values of transactions at agents prices for domestic consumption are the values at market prices plus any sales/purchase taxes, while for exports the values at world prices are the values at market prices less any export taxes.

The GAMS programmes that carry out the transformation of the GTAP database into a global SAM format are freely available, but are supplied without warranty. The programmes include a number of checks to ensure data consistency, but users are strongly encouraged to carry out checks for themselves.<sup>3</sup> No GTAP data are provided; it is assumed that those wishing to use these programmes are licensed to use the GTAP (version 5) database and have the GTAPAgg programme. The current versions of these programmes have only been tested with versions 5.0 and 5.3 of the GTAP database, and require that the GAMS user has at least

GAMS 20.7 complete with the GDX facilities installed; all the GAMS output are stored in GDX format. Users also need a copy of SEEHAR for converting the HAR file into a GAMS readable input file or copies of HAR2GAMS and GAMS2HAR.

The rest of this paper is organised as follows. The next section provides a description of the Social Accounting Matrix. The pertinent aspects of the data provided by the GTAP database are discussed in section 3, while section 4 provides a description of how the matrices of data from the GTAP database are used to construct the SAM. There is also an appendix that contains an extract from the GAMS programme that relates to the construction of the SAM. The SAM produced by the programme does not balance exactly; since this is a necessary condition for a ‘true’ SAM section 5 reports the magnitudes and relative importance of differences in the account totals for both SAMs, which are very minor, and then reports the steps taken to ensure balanced SAMs are produced. The maximum and minimum transactions values for each version of the database by region are reported in section 7, while section 8 offers some concluding comments.

## **2. Social Accounting Matrix**

The structure of the Global SAM is illustrated by means of a representative SAM for one region – the structure of the SAM for each and every region is identical. The description of the SAM structure is limited, since it is assumed that the reader is familiar with the basic principals of a SAM (see King, 1985; Sado ulet and de Janvry, 1995; Pyatt and Round, 1977; Pyatt, 1991; and Reinert and Roland-Holst, 1997). In general terms the SAM structure follows the conventions of the System of National Accounts for 1993 (UN, 1993), with adjustments in light of the limited data on intra-institutional accounts. The basic structure is identified in Table 2, where the active sub matrices of the SAM are identified with the dimensions of each active sub-matrix specified while in Table 3 the GTAP parameters that map onto the respective sub-matrices are identified. The SAM layout has an implicit number code for each sub-matrix given by the row and column numbers; this system of code numbers is used throughout the text to identify the sub-matrix currently being referred to.

The dimensions of the SAM are determined by accounts that are identified. The GTAP database identifies 57 sectors, where the term sector is used interchangeably to define

---

<sup>3</sup> If errors are uncovered please let the authors know; preferably with a copy of the code used to generate

imported and domestic produced commodities and production activities. Hence it is necessary to define 114 commodity accounts (57 for imported and 57 for domestically produced commodities) and 57 activity accounts. The database allows each domestic agent to be charged different purchase, or sales, tax rates for both imported and domestic commodities; hence it is necessary to have 114 accounts for sales taxes. Similarly each production activity can be charged an activity specific indirect/production tax, which necessitates another 57 accounts. Production also involves the use of primary inputs, of which 5 are identified in GTAP; so there are five accounts for factor purchases and five accounts for taxes paid on factor use.

For trade relations, each region **can** import from and export to all other regions,<sup>4</sup> hence there needs to be one account for each region, i.e., 66 in version 5.0 and 78 in version 5.3. In addition there needs to be two accounts for each region that a region **can** trade with to capture trade taxes – import duties and export taxes. And since there are three transport margins associated with each trade transaction there needs to be three accounts for each region that a region **can** trade with to capture trade and transport costs; with a further three accounts to capture exports by each region to the global trade and transport pool. Finally there are four domestic institutional accounts: the regional household that gathers together all regional income, either from factor sales or taxes (indirect and direct). This income is dispersed to the other three domestic institutions.

Consequently the dimensions of the SAM can be defined in general terms by reference to the number of commodities,  $c$ ; activities,  $a$ ; factors,  $f$ , and the number of regions,  $k$ . These are defined in Table 1.

**Table 1**                      **SAM Dimensions**

<b>Description</b>	<b>Code</b>	<b>Version 5.0</b>	<b>Version 5.3</b>
Commodities	$c$	57	57
Activities	$a$	57	57
Factors	$f$	5	5
Regions	$k$	66	78
Domestic Institutions	none	4	4

---

<sup>4</sup> and detect the errors.  
Strictly each region can be assumed not to trade with itself; hence it would be possible to exclude an account in each regional SAM.

The most immediately obvious points about the SAM are the large number of accounts and the relative scarcity of entries in the SAM. The large size of the SAM is primarily a consequence of the number of regions,  $k$ , in the database ( $k$  trade accounts plus  $2k$  for trade taxes) and the fact that 3 types of transport margins (i.e. air, water, land) on imports are identified with imports from each region ( $3k$  accounts), while the second main reason is the series of commodity/user specific sales taxes on imports ( $c$ , where  $c$  is the number of commodities) and domestic commodities ( $c$ ).

The resultant SAM is of the dimensions  $(4c + a + 2f + 9 + 6k) * (4c + a + 2f + 9 + 6k) * (k)$ , where  $c$  and  $a$  are the number of sectors in the database,  $f$  the number of factors and  $k$  the number of regions. Hence for version 5.0 the resultant SAM is of the dimensions  $701 * 701 * 66$  and for version 5.3 the dimensions are  $773 * 773 * 78$ .

The dimensions of the SAM indicate several very important features of the GTAP database.

- Information is concentrated in the trade accounts.
- The within regional information emphasises inter-industry and final demand transactions.
- The tax information relates overwhelmingly to indirect taxes on sales and purchases.
- There is very little information about domestic institutions other than for consumption.

**Table 2 Dimensions of the Social Accounting Matrix for a Representative Region**

		1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18
		Imported Commod	Domestic Commod	Activities	Factors	Regional Household	Private Household	Trade Taxes	Import Sales Taxes	Domestic Sales Tax	Factor Taxes	Production Taxes	Direct Taxes	Govt	Capital	Import Margins	Export Margins	Rest of World	Totals
1	Imported Commodities			<i>c*a</i>			<i>c*1</i>							<i>c*1</i>	<i>c*1</i>				<i>c*1</i>
2	Domestic Commodities			<i>c*a</i>			<i>c*1</i>	<i>c*c</i>						<i>c*1</i>	<i>c*1</i>		<i>c*3</i>	<i>c*k</i>	<i>c*1</i>
3	Activities		<i>a*c</i>																<i>c*1</i>
4	Factors			<i>f*a</i>															<i>f*1</i>
5	Regional Household				<i>1*f</i>			<i>1*132</i>	<i>1*c</i>	<i>1*c</i>	<i>1*5</i>	<i>1*1</i>	<i>1*1</i>						<i>1*1</i>
6	Private Household					<i>1*1</i>													<i>1*1</i>
7	Trade Taxes	<i>2k*c</i>	<i>2k*c</i>																<i>2k*1</i>
8	Import Sales Taxes			<i>c*a</i>			<i>c*1</i>							<i>c*1</i>	<i>c*1</i>				<i>c*1</i>
9	Domestic Sales Tax			<i>c*a</i>			<i>c*1</i>							<i>c*1</i>	<i>c*1</i>				<i>c*1</i>
10	Factor Taxes			<i>f*a</i>															<i>f*1</i>
11	Production Taxes			<i>1*a</i>															<i>1*1</i>
12	Direct Taxes				<i>1*f</i>														<i>1*1</i>
13	Government					<i>1*1</i>													<i>1*1</i>
14	Capital				<i>1*f</i>	<i>1*1</i>												<i>1*3</i>	<i>1*k</i>
15	Import Margins	<i>3k*c</i>																	<i>3k*1</i>
16	Export Margins															<i>3*3k</i>			<i>3*1</i>
17	Rest of World	<i>k*c</i>																	<i>k*1</i>
18	Totals	<i>1*c</i>	<i>1*c</i>	<i>1*a</i>	<i>1*f</i>	<i>1*1</i>	<i>1*1</i>	<i>1*2k</i>	<i>1*c</i>	<i>1*c</i>	<i>1*f</i>	<i>1*1</i>	<i>1*1</i>	<i>1*1</i>	<i>1*1</i>	<i>1*3k</i>	<i>1*3</i>	<i>1*k</i>	



**Table 3 Transactions in the Social Accounting Matrix for a Representative Region**

		1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17
		Imported Commod	Domestic Commod	Activities	Factors	Regional Household	Private Household	Trade Taxes	Import Sales Taxes	Domestic Sales Tax	Factor Taxes	Production Taxes	Direct Taxes	Govt	Capital	Import Margins	Export Margins	Rest of World
1	Imported Commodities			Intermediate Inputs			Private demand							Government demand	Investment demand			
2	Domestic Commodities			Intermediate inputs			Private demand							Government demand	Investment demand		Exports of Transport Services	Exports of Goods and Services
3	Activities		Supply matrix															
4	Factors			Payments to Factors														
5	Regional Household				Distributed Factor Incomes			Import Duties and Export duties	Import Sales Taxes	Domestic Sales Taxes	Factor Use Taxes	Production Taxes	Income Taxes					
6	Private Household					Private Household income												
7	Trade Taxes	Import Duties	Export duties															
8	Import Sales Taxes			Sales Taxes - Imports			Sales Taxes - Imports							Sales Taxes - Imports	Sales Taxes - Imports			
9	Domestic Sales Tax			Sales Taxes - Domestic Factor Use Taxes			Sales Taxes - Domestic							Sales Taxes - Domestic	Sales Taxes - Domestic			
10	Factor Taxes																	
11	Production Taxes			Production Taxes														
12	Direct Taxes				Factor Income Taxes													
13	Government					Government income												
14	Capital				Depreciation	Savings											Trade balance	Trade balance
15	Import Margins	Transport margins on imports																
16	Export Margins																Imports of transport services	
17	Rest of World	Imports of Goods and Services																

## SAM Transactions

The SAM reports six groups of agents that demand commodities (reading across the rows of the commodity accounts); activities, private households, government, investment, global transport services and other regions. These transactions take place at sellers' prices, which GTAP terms market prices. However, associated with each purchase by an agent there is an additional payment to the government that represents sales taxes. The SAM demonstrates that for imported commodities sellers prices are the price received by the exporting region plus the per unit transport cost plus the per unit tariff rate (reading down the columns of the commodity accounts), while for domestically produced commodities sellers prices are the (producer) prices received by domestic activities; note how export taxes are recorded. Domestic producer prices are derived from the production costs, which are made up of the costs of intermediate inputs valued at sellers' prices plus the sales taxes paid on intermediate inputs, plus the expenditure on factor use and production taxes and payments to primary inputs.

All income initially accrues to the regional household before being distributed to the private household, government and the capital account. Regional income is made up of payments to factors, net of factor income taxes and depreciation, plus indirect taxes, which are made up of trade, sales, factor use and production taxes, and income taxes. The artifice of a regional household means that the private household does not pay income taxes nor does it save, and the government neither saves nor borrows, i.e., the balance on the government budget is zero.

Trade consists of two elements; expenditures on commodities and expenditures on transport margins. Exports are valued 'free on board' and after the payment of any export duties. Exports of transport services to the global transport pool are recorded separately. Imports of commodities are also valued fob, with transport services recorded separately. The sum of the two represents expenditure on imports inclusive of carriage, insurance and freight (cif). Consequently there are two sets of trade balances. The first represents the trade balance with respect to each of the transport service while the second is the trade balance with each and every region.

### 3 The GTAP Data

It is useful to start with some brief notes about the GTAP 5 database and, since recourse will be made to extracts of GEMPACK code, some aspects of GEMPACK. These notes fall into 5 categories; modeling in terms of rates of change, sets and the local control of sets, the regional household, trade and transport margins and the global bank. As a matter of convention, in the GTAP model all variables, i.e., the rates of change, are in lower case and the parameters are in upper case. Subsequently this section goes on to identify the data from the GTAP database that are used to define transactions.

#### *Rates of Change*

One feature of GEMPACK that might cause discomfiture to GAMS modelers is the solution of models in terms of rates of changes – in essence the models are written in terms of log differentials. One area in which this can cause some confusion is the treatment of taxes. Typically, ‘levels modelers’ might be inclined to talk about the *tax rates*, whereas ‘rates of changes modelers’ might talk about the *powers of taxes*. The explanation is straightforward. Consider the level relationship for an *ad valorem* tax

$$P_1 = P_2 * (1 + T)$$

where  $T$  is the *ad valorem* tax rate and  $(1 + T)$  is defined as the **power of the tax**. In terms of rates of change this can be written as

$$d \ln(P_1) = d \ln(P_2) + d \ln(1 + T)$$
$$p_1 = p_2 + t$$

such that  $t$  is the proportionate change in the power of the tax. It is useful to note this when reading the GTAP code for the tax system.

#### *SETS and Control of SETS*

The GTAP model – gtap61.tab (GTAP, 2002; Hertel, 1997) - defines the following sets that are relevant when transforming GTAP data into a SAM format.

```
REG # regions in the model #
TRAD_COMM # traded commodities #
MARG_COMM # margin commodities #
NMRG_COMM # non-margin commodities # = TRAD_COMM - MARG_COMM;';
CGDS_COMM # capital goods commodities #
ENDW_COMM # endowment commodities #
PROD_COMM # produced commodities # = TRAD_COMM union CGDS_COMM;
DEMD_COMM # demanded commodities # = ENDW_COMM union TRAD_COMM;
NSAV_COMM # non-savings commodities # = DEMD_COMM union CGDS_COMM;
ENDWS_COMM
# sluggish endowment commodities # = (all,i,ENDW_COMM: SLUG(i)>0);
ENDWM_COMM
```

```
# mobile endowment commodities # = ENDW_COMM - ENDWS_COMM;
ENDWC_COMM # capital endowment commodity # (capital);
```

The GTAP model also contains a useful illustration of the relationships between the sets and subsets, an adaptation of which is presented below.

NSAV_COMM			
DEMD_COMM			
	PROD_COMM		
ENDW_COMM	TRAD_COMM		CGDS_COMM
	MARG_COMM	NMARG_COMM	
ENDWM_COMM	ENDWS_COMM		

It is important, for GAMS users, to note that when reading the extracts of GTAP model code reproduced below that the model is coded in GEMPACK where the sets are local – hence the inclusion of local definitions for variables and parameters. To understand the local control of sets consider the following extract of code from the GTAP tablo<sup>5</sup> file

```
Coefficient (all,i,DEMD_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFA(i,j,r) # producer expenditure on i by j in r valued at agents' prices #;
Formula (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFA(i,j,r) = EVFA(i,j,r);
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
  VFA(i,j,s) = VDFa(i,j,s) + VIFA(i,j,s);
```

This first defines the coefficient (parameter in GAMS) VFA over  $i$ ,  $j$  and  $r$  with members of  $i$  defined as DEMD\_COM, of  $j$  as PROD\_COM and of  $r$  as REG. The subsequent two formulae are then constrained to operate over subsets of  $i$ , i.e., ENDW\_COM and TRAD\_COM (see the schematic above), but they can reuse the character  $i$  because of the system of local control of sets. Similarly it is perfectly acceptable to redefine  $i$  for subsequent use.

A major consequence of this feature of GEMPACK is the necessity to explicitly identify the set definitions that apply to any particular block of GEMPACK code.

### *Regional Household*

The GTAP database and model use the concept of a Regional Household. Since this is crucial to an understanding of the relationships between transactions it is useful to provide a brief description of the Regional Household concept here. In the GTAP database all incomes from indirect taxes and the (limited) direct taxes on factor incomes, and all factor incomes accrue to the Regional Household. These domestically sourced incomes are then distributed between the Private Household, the Government and the Capital (saving-investment) accounts. Hence

<sup>5</sup> In GEMPACK a tablo file is the name given to a programme file – the programme for editing these files is known as TABMATE.

the Regional Household is the sole source of income to both the private household and the government, although there are two other sources of income for the capital account, namely depreciation and the balance on the capital account of the balance of payments. All three domestic institutions are then limited to only making transactions with the commodity accounts, gross of any sales taxes. Consequently, the private household does not record transactions directly with either the government or the capital account and the government account does not record transactions with the capital account; hence the private household neither saves nor pays income taxes and the government does not run a surplus or deficit on its current account.

In essence therefore the Regional Household provides a method by which the limitations imposed by the absence of data on intra-institutional transactions can be partially circumvented. This method does however have a cost in modeling terms; it involves the imposition of a particular vision on the distribution of consumption expenditure by domestic institutions.

#### *Trade and Transport Margins*

The GTAP database records the transactions with respect to three types of trade and transport margin for each commodity imported by each region from each region. The sum of these margins for each commodity imported from each region are defined as the difference between the fob price of the commodity exported from region  $x$  to region  $y$  and the cif price of each commodity imported by region  $y$  from region  $x$ . But the database does not record data on the regions that supply each trade and transport margins transaction, i.e., there is not full information on bilateral trade in trade and transport margins. Instead the database records the total value of trade and transport margin exports by each region; the global sum of the values of the exports of each trade and transport margin must equal the global sum of the values of the respective imports of trade and transport margins. But it is not necessary that the values of the exports of trade and transport margins by each region must equal the values of the imports of trade and transport margins by each region.

In the GTAP model this is handled by the concept of a global pool/bank for trade and transport margins. This pool supplies all the demands for (the import of) trade and transport margins and then clears the global market for trade and transport margins by purchasing the margins from the regions that export trade and transport margins. Hence each region runs a trade account with the global pool for trade and transport margins, which must *ex post* be in

balance. Since the total values of margin exports and imports for each region and each margin goods are known, the balance in trade and transport margins trade can be readily calculated. In database terms this is not a problem, although issues may remain about the appropriate modeling approach.

### *The Global Bank*

A related issue exists with respect to global savings. The total value of global savings must equal the total value of global investments and the values of total savings and investment within each region must be equal for all regions. But total savings by domestic institutions within each region need not equal the total value of investment within each region because of inter regional capital account transactions. Furthermore, the *ex post* the balance of payments for each region must balance and the global sum of the balances of trade in goods and services must equal zero, as must the global sum of inter regional capital account transactions.

This presents economic logic problems for experiments since the logic of a price driven model suggests that differences in investment expenditures across regions would be influenced by differences in returns to capital in different regions. Nevertheless this is not a substantive issue for the reporting of transactions in the database since it simply records transactions in the base period and by definition these must satisfy the requisite accounting conditions. Hence for database purposes it is sufficient to record the calculated balance of trade in goods and services for each region with every other region. In database terms this is not a problem, although issues may remain about the appropriate modeling approach.

### Data Read in by GTAP Model

The GTAP model only uses a subset of the GTAP database to define the transactions within and between regions in the base period. Consequently it is only necessary to use that subset in the derivation of the global SAM. Details of the data read in by the GTAP model are reported, in terms used in the GTAP model, below.

GTAP has a generally consistent convention for labeling data parameters/coefficients, which, while not applying to all data matrices, is extremely useful in reading the code.

- V Value;
- D Domestic;
- I Imported;
- P Private household;

- G Government;
- F Firms;
- A Agents prices;
- M Market prices;
- E Endowment;
- X Exports;
- W World.

Hence the data matrix VIGA is the Value of Imports for Government consumption at Agents prices.

The data are:

*Variables and base flows associated with saving*

Coefficient (all,r,REG)  
 SAVE(r) # expenditure on NET savings in region r valued at agent's prices

*Variables and base flows associated with government consumption*

Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VDGA(i,r) # government consumption expenditure on domestic i in r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VDGM(i,r) # government consumption expenditure on domestic i in r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VIGA(i,r) # government consumption expenditure on imported i #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VIGM(i,r) # gov't consumption expenditure on i in r #;

*Variables and base flows associated with private consumption*

Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VDPA(i,r) # private consumption expenditure on domestic i in r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VDPM(i,r) # private consumption expenditure on domestic i in r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VIPA(i,r) # private consumption expenditure on imported i in r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,r,REG)  
 VIPM(i,r) # private consumption expenditure on i in r #;

*Variables and base flows associated with the firms*

Coefficient (ge 0)(all,i,ENDW\_COMM)(all,r,REG)  
 EVOA(i,r) # value of commodity i output in region r #;  
 Coefficient (ge 0)(all,i,ENDW\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 EVFA(i,j,r) # producer expenditure on i by j in r valued at agent's prices#;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 VDFA(i,j,r) # purchases of domestic i for use by j in region r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 VIFA(i,j,r) # purchases of imported i for use by j in region r #;  
 Coefficient (ge 0)(all,i,ENDW\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 VFM(i,j,r) # producer expenditure on i by j in r valued at mkt prices #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 VIFM(i,j,r) # purchases of imports i for use by j in region r #;  
 Coefficient (ge 0)(all,i,TRAD\_COMM)(all,j,PROD\_COMM)(all,r,REG)  
 VDFM(i,j,r) # purchases of domestic i for use by j in region r #;

Note that the set definitions for these parameters, excluding EVOA, are defined over *PROD\_COMM*. This is because investment expenditure is treated as a demand by firms for capital commodities (*CGDS*). This raises a minor complication through the need to separate *CGDS* out to derive investment demand and results in the declaration of a tax parameter that is **always** assigned a zero value.

#### *Variables and base flows associated with the global bank*

```
Coefficient (ge 0)(all, r, REG)
  VDEP(r) # value of capital depreciation in r (exogenous) #;
  Value of capital depreciation is used to compute the (exogenous) depreciation
  rate and therefore does not appear in update.
```

#### *Variables and base flows associated with the international trade and transport*

```
Coefficient (ge 0)(all, i, TRAD_COMM)(all, r, REG)(all, s, REG)
  VIMS(i, r, s) # imports of i from r to s valued at domestic mkt prices #;
Coefficient (ge 0)(all, i, TRAD_COMM)(all, r, REG)(all, s, REG)
  VIWS(i, r, s) # imports of i from r to s valued cif (tradeables only) #;
Coefficient (ge 0)(all, i, TRAD_COMM)(all, r, REG)(all, s, REG)
  VXWD(i, r, s) # exports of i from r to s valued fob (tradeables only) #;
Coefficient (ge 0)(all, i, TRAD_COMM)(all, r, REG)(all, s, REG)
  VXMD(i, r, s) # exports of i from r to s valued at mkt prices (tradeables only)
  #;
Coefficient (ge 0)(all, m, MARG_COMM)(all, r, REG)
  VST(m, r) # exports of m from r for int'l trnsport valued at mkt p (tradeables
  only) #;
```

## **4 Constructing the Social Accounting Matrix**

This section describes how the various sub matrices of the SAM are assigned. Table 4 presents a map of the assignments where, in the main, the notation is derived from the GTAP data labels. The major issues all relate to the treatment of taxes and hence to the systems by which prices are formed. The part of the GAMS programme that carries out these actions is listed in Appendix

The description of the assignment statements starts with the components of domestic demand, and domestic production. The discussion then turns to government income which involves specifying the tax transactions. The discussion then moves onto the purchases of inputs by activities and then to the components of regional income. Finally the discussion turns to the treatment of trade and the associated trade and transport margins and thence to income to the capital account, i.e., the balances on the capital account with respect to the rest of the world.



## GAMS Sets

The sets used in the GAMS code, where SS is the global set, are

### Commodity sets

MCC(SS)	Imported Commodities
DCC(SS)	Domestic Commodities

### Activity sets

AA(SS)	Activities
A2	Production commodity Activities in GTAP

### Factor sets

FF (SS)	Factors
---------	---------

### Tax sets

TMMTAX(SS)	Import duties
TEETAX(SS)	Export taxes
SMMTAX(SS)	Sales Taxes on imported goods and services
SDDTAX(SS)	Sales Taxes on domestic goods and services
FFTAX(SS)	Taxes on Factors

### Trade sets

OTPPMARG(SS)	Other trade and transport costs
WTPPMARG(SS)	Water trade and transport costs
ATPPMARG(SS)	Air trade and transport costs
VSTTMARG(SS)	Exports of trade and transport services
WW(SS)	Rest of the World
KK	Regions

In addition a set A2 is also declared, but it is not necessary to declare this as a subset of SS.

This set corresponds to the set TRAD\_COMM in the GTAP model since it includes all commodities that are potentially traded AND the set CGDS\_COM, and is only used partition the demand for intermediate goods from the demand for investment goods.

**Table 4 Mapping Transactions: Social Accounting Matrix for a Representative Country**

		1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18
		Imported Commod	Domestic Commod	Activities	Factors	Regional Household	Private Household	Trade Taxes	Import Sales Taxes	Domestic Sales Tax	Factor Taxes	Production Taxes	Direct Taxes	Govt	Capital	Import Margins	Export Margins	Rest of World	Totals
1	Imported Commodities			VIAM			VIPM							VIGM	VIIM				1:18
2	Domestic Commodities			VDAM			VDPM							VDGM	VDIM		VST	VXWD	2:18
3	Activities		VOM																3:18
4	Factors			VFM															4:18
5	Regional Household				EVOA-VDEP			TMTAX + TETAX	ISTAX	DSTAX	FTAX	PTAX	DTAX						5:18
6	Private Household					YH													6:18
7	Trade Taxes	VIMS- VIWS	VXMD - VXWD																TMTAX TETAX
8	Import Sales Taxes			VIAA-VIAM			VIPA-VIPM							VIGM-VIGA	VIIA-VIIM				ISTAX
9	Domestic Sales Tax			VDAA- VDAM			VDPA- VDPM							VDGM- VDGA	VDIA- VDIM				DSTAX
10	Factor Taxes			EVFA-VFM															FTAX
11	Production Taxes			PRODTAX											PTAXINV				PTAX
12	Direct Taxes				PTAXFACT														DTAX
13	Government					YG													13:18
14	Capital																		
15	Import Margins				VDEP	SAVE											VTWR- VST	VTWR- VXWD	14:18
16	Export Margins	VTWR																	15:18
17	Rest of World	VIWS- VTWR														VTWR			16:18
18	Totals	18:1	18:2	18:3	18:4	18:5	18:6	18:7	18:8	18:9	18:10	18:11	18:12	18:13	18:14	18:15	18:16	18:17	17:18

A naming convention is adopted in the GAMS code for each GTAP parameter/coefficient used in the programme. All the GTAP parameters are read in using the names and sets assigned to them in the HAR file. But the local control of sets within GEMPACK means that it is possible for a set, e.g., *i*, to be used with multiple definitions, e.g., for commodities and activities, which goes against the logic of the SAM account structure and the global control of sets in GAMS. Hence as part of the transition process it is necessary to re label the accounts for each parameter. To ease readership of the code any re labeling of a GTAP parameter produces a new parameter with the same name plus a number, e.g., VXMD(TRAD\_COMM,REG,REG) becomes VXMD1(dcc,kk,ww); in some cases multiple versions are required to complete parts of the global SAM, e.g., VXMD2(dcc,kk,teetax).

There is one substantive exception to this general rule. Demand for investment goods demand is recorded in the same parameter as the demand for intermediate inputs and it convenient to separate out these transactions. Hence VDFA(TRAD\_COMM,PROD\_COMM,REG) becomes VDFA1(dcc,a2,kk),<sup>6</sup> which is subsequently transformed into VDAA1(dcc,aa,kk) and VDIA1(dcc,ii,kk), where the ‘F’ becomes ‘A’, for activities, and ‘I’ for investment.

### Commodity Demand by Domestic Agents

Commodity demand by domestic agents consists of final demand by domestic institutions – private household, government and investment – and intermediate demand by activities. Domestic demand is further disaggregated by distinguishing between imported and domestically produced commodities. If standard national accounting conventions for a SAM are followed, purchases by domestic agents would be valued at purchaser prices (agents prices in the GTAP database), but, for reasons to do with the indirect tax instruments that are discussed below, these transactions needed to be recorded at sellers prices (market prices in the GTAP database).

There are two key prices for the commodity demand system. The market price of the domestic commodity *i* in region *r*,  $pm_{i,r}$ , and market price of the imported commodity *i* in region *r*,  $pim_{i,r}$ . The relevant GTAP code is given below.

```
Variable (orig_level = 1.0)(all,i,NSAV_COMM)(all,r,REG)
    pm(i,r) # market price of commodity i in region r #;
```

---

<sup>6</sup> Note the use of the set A2 in VDFA1.

```
Variable (orig_level = 1.0)(all,i,TRAD_COMM)(all,r,REG)
  pim(i,r) # market price of composite import i in region r #;
```

The market price is defined as the price received by the seller of the commodity. In a SAM that follows the SNA conventions for supply and use tables the commodity demands by domestic agents would be valued in purchaser prices; however that convention requires that a law of one price holds. Hence for this SAM the domestic commodity demands by domestic agents are valued in seller prices, which are uniform for all purchasers. (See the notes on the tax system below for an explanation of how this affects the recording of taxes.)

### *Intermediate Demand and Investment (1:3, 2:3, 1:14, 2:14)*

Intermediate demands for imported and domestic commodities are components of purchases by firms. In market price terms these are identified in the GTAP tablo file as follows

```
Coefficient (ge 0)(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
  VIFM(i,j,r) # purchases of imports i for use by j in region r #;
Update (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
  VIFM(i,j,r) = pim(i,r) * qfm(i,j,r);
Read
  VIFM from file GTAPDATA header "VIFM";

Coefficient (ge 0)(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
  VDFM(i,j,r) # purchases of domestic i for use by j in region r #;
Update (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
  VDFM(i,j,r) = pm(i,r) * qfd(i,j,r);
Read
  VDFM from file GTAPDATA header "VDFM";
```

where

```
Variable (orig_level=VDFM)(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
  qfd(i,j,s) # domestic good i demanded by industry j in region s #;
Variable (orig_level=VIFM)(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
  qfm(i,j,s) # demand for i by industry j in region s #;
```

Since the SAM separately records demands by activities (intermediate demand) and investment demand the GAMS code creates new parameters to carry the intermediate and investment demand data.

### *Private Household Demand (1:6, 2:6)*

The demand for domestic and imported commodities by the (domestic) private households, one per region, are identified in the GTAP tablo file in market price values as

```
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)
  VDPM(i,r) # private consumption expenditure on domestic i in r #;
Update (all,i,TRAD_COMM)(all,r,REG)
  VDPM(i,r) = pm(i,r) * qpd(i,r);
Read
  VDPM from file GTAPDATA header "VDPM";

Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)
  VIPM(i,r) # private consumption expenditure on i in r #;
Update (all,i,TRAD_COMM)(all,r,REG)
  VIPM(i,r) = pim(i,r) * qpm(i,r);
Read
  VIPM from file GTAPDATA header "VIPM";
```

where

```
Variable (orig_level=VDPM)(all,i,TRAD_COMM)(all,s,REG)
    qpd(i,s) # private hhld demand for domestic i in region s #;
Variable (orig_level=VIPM)(all,i,TRAD_COMM)(all,s,REG)
    qpm(i,s) # private hhld demand for imports of i in region s #;
```

### *Government Demand (1:13, 2:13)*

The demand for domestic and imported commodities by the regional governments are identified in the GTAP tablo file in market price values as

```
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)
    VDGM(i,r) # government consumption expenditure on domestic i in r #;
Update (all,i,TRAD_COMM)(all,r,REG)
    VDGM(i,r) = pm(i,r) * qgd(i,r);
Read
    VDGM from file GTAPDATA header "VDGM";

Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)
    VIGM(i,r) # gov't consumption expenditure on i in r #;
Update (all,i,TRAD_COMM)(all,r,REG)
    VIGM(i,r) = pim(i,r) * qgm(i,r);
Read
    VIGM from file GTAPDATA header "VIGM";
```

where

```
Variable (orig_level=VDGM)(all,i,TRAD_COMM)(all,s,REG)
    qgd(i,s) # government hhld demand for domestic i in region s #;
Variable (orig_level=VIGM)(all,i,TRAD_COMM)(all,s,REG)
    qgm(i,s) # government hhld demand for imports of i in region s #;
```

### Domestic Commodity Output (3:2)

Domestic output for the domestic market (*VDM*) is defined as the sales on the domestic market of commodities produced by domestic activities. In the GTAP tablo file this is given as

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) # domestic sales of i in r at mkt prices (tradeables only) #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) = VDPM(i,r) + VDGM(i,r) + sum(j,PROD_COMM, VDFM(i,j,r));
```

Hence

$$VDM_{dc,k} = VDPM_{dc,k} + VDGM_{dc,k} + VDAM_{dc,k} + VDIM_{dc,k}$$

But the convention adopted for this SAM is that exports are from the domestic commodity accounts.<sup>7</sup> Hence the supply matrix is defined as the total value of domestic production destined for domestic and export (foreign) markets. Hence total domestic supply is defined as VDM plus exports, commodity and transport services, and export taxes. This is defined as part of VOM in the GTAP tablo file, i.e.,

---

<sup>7</sup> This contrasts with the SAM layout implicit to the GTAP model, which follows the practice of treating exports as originating from the activity accounts (for an early example of this layout see Dervis *et al.*, 1984). Either arrangement is acceptable with the reduced form input-output table used in the GTAP database.

```

Coefficient (all,i,NSAV_COMM)(all,r,REG)
  VOM(i,r) # value of commodity i output in region r at market prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
  VOM(i,r) = sum(j,PROD_COMM, VFM(i,j,r));
Formula (all,m,MARG_COMM)(all,r,REG)
  VOM(m,r) = VDM(m,r) + sum(s,REG, VXMD(m,r,s)) + VST(m,r);
Formula (all,i,NMRG_COMM)(all,r,REG)
  VOM(i,r) = VDM(i,r) + sum(s,REG, VXMD(i,r,s));
Formula (all,h,CGDS_COMM)(all,r,REG)
  VOM(h,r) = VOA(h,r);

```

although the presentation is complicated by the definition of *VOM* to include the values of endowments and capital goods at market prices.

Hence

$$\begin{aligned}
 VOM_{dc,k} &= VDM_{dc,k} + VXMD_{dc,k} + VST_{m,k} \\
 &= VDM_{dc,k} + VXWD_{dc,k} + ETAX_{dc,k} + VST_{m,k}
 \end{aligned}$$

### Expenditures by Activities/Incomes to Factors

Expenditures by activities fall under three headings: purchases of intermediate inputs, primary inputs and payment of indirect taxes. Purchases of intermediate inputs are explained as aspects of commodity demand. These are valued at market prices with the associated ‘sales’ taxes being treated as separate expenditures by the activities – see the section on indirect taxes. The treatment of production taxes (PRODTAX) is also discussed in the section on indirect taxes.

This leaves the payments by activities for factor services (4:3). The GTAP tablo file indicates that the payments to factor accounts for factor services are defined as

```

Coefficient (all,i,NSAV_COMM)(all,r,REG)
  VOM(i,r) # value of commodity i output in region r at market prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
  VOM(i,r) = sum(j,PROD_COMM, VFM(i,j,r));

```

where

```

Variable (all,i,ENDWS_COMM)(all,j,PROD_COMM)(all,r,REG)
  pmes(i,j,r) # market price of sluggish endowment i used by j, in r #;
Coefficient (ge 0)(all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) # producer expenditure on i by j in r valued at mkt prices #;
Update (all,i,ENDWM_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) = pm(i,r) * qfe(i,j,r);
Update (all,i,ENDWS_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) = pmes(i,j,r) * qfe(i,j,r);
Read
  VFM from file GTAPDATA header "VFM";

```

Hence they are valued in market price terms. This means that there is a tax on the use of factors that is activity specific – this is explained below in the section on indirect taxes.

Sales of factor services to domestic activities are the sole sources of factor incomes in the database.

## Factor Expenditures

Factor incomes are distributed to domestic agents in three sub matrices. The first is the allowance for depreciation (14:4)

```
Variable (all, r, REG)
  kb(r) # beginning-of-period capital stock, in r #;
Variable (all, r, REG)
  pcgds(r) # price of investment goods = ps("cgds",r) #;
Coefficient (ge 0)(all, r, REG)
  VDEP(r) # value of capital depreciation in r (exogenous) #;
Update (all, r, REG)
  VDEP(r) = kb(r) * pcgds(r);
Read
  VDEP from file GTAPDATA header "VDEP";
```

which is recorded as an expenditure by the factors accounts to the domestic capital account, i.e., a form of saving.

Second the factor account must pay factor income taxes, which are defined as the totals of the payments to each factor by all activities less the ‘endowments output at agents’ prices (12:4). These are explained below.

And the third component is the distribution of factor income after allowance for depreciation and the payment of factor taxes to the regional household (5:4). Hence the factor incomes distributed to the regional household are the ‘endowments output at agents’ prices’

```
Variable (all,i,NSAV_COMM)(all,r,REG)
  ps(i,r) # supply price of commodity i in region r #;
Variable (orig_level=VOM)(all,i,NSAV_COMM)(all,r,REG)
  qo(i,r) # industry output of commodity i in region r #;
Coefficient (ge 0)(all,i,ENDW_COMM)(all,r,REG)
  EVOA(i,r) # value of commodity i output in region r #;
Update (all,i,ENDW_COMM)(all,r,REG)
  EVOA(i,r) = ps(i,r) * qo(i,r);
Read
  EVOA from file GTAPDATA header "EVOA";
```

less the allowance for depreciation.

## Regional Household Income

Regional household income consists of distributed factor incomes, which were explained above, and total tax revenues, which are defined as indirect tax income plus direct tax income. Because of the Regional Household system, which means that private households pay no direct/income taxes and the private household and government do not save, the tax revenues in GTAP are dominated by indirect taxes.

The definitions for these taxes are crucial to an accurate accounting of transactions in the GTAP database. The definitions for the tax receipts in the GTAP database are taken from

version 6.1 of the GTAP model tablo file, as set out in the section ‘Computation of Indirect Tax Receipts by the Regional Household’.<sup>8</sup>

### *Indirect Tax Revenues*

Indirect tax receipts are defined in the GTAP model tablo file (gtab.tab v6.1) as

```
INDTAX(r) # indirect tax receipts in r #;
INDTAX(r) =
TPC(r) + TGC(r) + TIU(r) + TFU(r) + TOUT(r) + TEX(r) + TIM(r);
```

where

```
TIM(r) # import tax payments in r #;
TEX(r) # export tax payments in r #;
TIU(r) # firms' tax payments on intermediate goods usage in r #;
TGC(r) # government consumption tax payments in r #;
TPC(r) # private consumption tax payments in r #;
TOUT(r) # production tax payments in r #;
TFU(r) # firms' tax payments on primary factor usage in r #;
```

The definitions of each of these indirect tax instruments in the GTAP database and how they are treated in the global SAM are explained below. When considering the GTAP tax instruments it is important to note the convention in the GTAP database and model by which the tax revenue is defined; the right hand side elements of the tax revenue equations are ordered A(gents), M(arket), W(orld), and hence it is also important to note how prices are formed in the GTAP model.

### *Trade Taxes*

#### *Import Duties (7:1)*

The GTAP model defines the revenues from import duties (*MTAX*) on commodity *i* imported from region *r* by region *s* as

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
MTAX(i,r,s) # tax on imports of good i from source r in destination s #;
Formula (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
MTAX(i,r,s) = VIMS(i,r,s) - VIWS(i,r,s);
Coefficient (all,r,REG)
TIM(r) # import tax payments in r #;
Formula (all,r,REG)
TIM(r) = sum(i,TRAD_COMM,sum(s,REG, MTAX(i,s,r)));
```

Further the (rates of change in) market prices (*pms*) are defined as

```
Equation MKTPRICES
# eq'n links domestic and world prices (HT 24) #
(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
pms(i,r,s) = tm(i,s) + tms(i,r,s) + pcif(i,r,s);
```

where *tms* is the import duty rate and *pcif* is the cif paid import price (*tm* is a variable included for purposes of simulations and in the current context is not relevant). Hence import

<sup>8</sup> This starts at line 747 in gtap61.tab.



duty revenues must be added to the payments to the source region to identify the payments for imports made by domestic consumers.

### *Export Taxes (7:2)*

Export taxes are defined as the revenues from taxes charged on exports (*ETAX*) on commodity *i* exported from region *r* to region *s*

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  XTAXD(i,r,s) # tax on exports of good i from source r to destination s #;
Formula (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  XTAXD(i,r,s) = VXWD(i,r,s) - VXMD(i,r,s);
Coefficient (all,r,REG)
  TEX(r) # export tax payments in r #;
Formula (all,r,REG)
  TEX(r) = sum(i,TRAD_COMM, sum(s,REG, XTAXD(i,r,s)));
```

The (rates of change in) export prices (*pfob*) are defined as

```
Equation EXPRICES
# eq'n links agent's and world prices (HT 27) #
(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  pfob(i,r,s) = pm(i,r) - tx(i,r) - txs(i,r,s);
```

where *txs* is the export tax rate, *pfob* the fob export price and *pm* the market price of commodity *i* in region *r*. Hence export tax revenues must be subtracted from the payments by the purchasing region to identify the revenues from exports received by domestic producers.

This is confirmed by the definitions of *VXMD*

```
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXMD(i,r,s)
  # exports of i from r to s valued at mkt prices (tradeables only) #;
Update (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXMD(i,r,s) = pm(i,r) * qxs(i,r,s);
```

and *VXWD*

```
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXWD(i,r,s) # exports of i from r to s valued fob (tradeables only) #;
Update (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXWD(i,r,s) = pfob(i,r,s) * qxs(i,r,s);
Read
  VXWD from file GTAPDATA header "VXWD";
```

Hence export taxes are defined by  $VXMD - VXWD$ .<sup>9</sup>

### *Taxes on Demand/Consumption*

There are 6/8 indirect taxes on demand – indirect taxes on demand for domestic and imported commodities by the private household, government, investment and activities, with the investment and activity demands being ‘lumped’ together under the label of firms’ demand.

<sup>9</sup> The convention followed in this SAM is that the definitions for prices are built up as accounting identities through the columns of the commodity accounts. Hence export taxes are accounted for as the difference between the price paid by the destination region and the price received by the domestic activity. The notable exception is the taxes on consumption.

Transactions between activities, the private household, government and investment and the commodity accounts are valued at the sellers' prices – which for these transactions are termed market prices by GTAP. Then transactions between the institutions and the sales tax accounts are recorded that identify the 'sales' taxes paid by each institution with respect to purchases of each commodity. Consequently sales tax transactions do not appear as row entries in the columns for the commodity accounts. While the purchasing institutions pay different prices the sellers receive a common price (*pm* or *pim*) irrespective of the institution that makes the purchase; hence the domestic supply matrix is recorded in market price terms (broadly equivalent to basic prices), while the supply of imported commodities is valued inclusive of trade taxes and transport margins for imported commodities, but excluding any domestic sales taxes.

The allocation method described below retains all the information in the GTAP database. Alternative representations of the GTAP database are consequently reduced forms.

*Intermediate and Investment Demand (8:3, 9:3, 8:14, 9:14)*

Taxes on purchases of intermediate and investment commodity differ according to whether the commodity is domestically produced or imported, i.e.,

```
Variable (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    tfd(i,j,r) # tax on domestic i purchased by j in r #;
Variable (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    tfm(i,j,r) # tax on imported i purchased by j in r #;
```

hence there are separate tax revenues for domestic and imported commodities, i.e.,

```
Coefficient (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    DFTAX(i,j,r) # tax on use of domestic intermediate good i by j in r #;
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    DFTAX(i,j,r) = VDFA(i,j,r) - VDFM(i,j,r);
Coefficient (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    IFTAX(i,j,r) # tax on use of imported intermediate good i by j in r #;
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    IFTAX(i,j,r) = VIFA(i,j,r) - VIFM(i,j,r);
Coefficient (all,r,REG)
    TIU(r) # firms' tax payments on intermediate goods usage in r #;
Formula (all,r,REG)
    TIU(r) = sum(j,PROD_COMM, sum(i,TRAD_COMM, DFTAX(i,j,r) + IFTAX(i,j,r)));
```

**But taxes are defined by**

```
Equation DMNDDPRICE
# eq'n links domestic market and firm prices (HT 20) #
(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    pfd(i,j,r) = tfd(i,j,r) + pm(i,r);
Equation DMNDIPRICES
# eq'n links domestic market and firm prices (HT 23) #
(all,i,TRAD_COMM)(all,j,PROD_COMM)(all,r,REG)
    pfm(i,j,r) = tfm(i,j,r) + pim(i,r);
```

where *pdf* and *pfm* define the prices paid by the 'activities' while *pm* and *pim* define the price received by the suppliers. Hence *VDFM* and *VIFM* record the transactions between the

commodity and ‘activity’ accounts and *DFTAX* and *IFTAX* define the associated tax payments by ‘activities’ to the government.

Note however that *VDFM/VFDA* and *VIFM/VIFA* are defined over *PROD\_COMM* and hence ‘activities’ include the investment account. In the GAMS mapping programme *VDFM/A* are split into *VDAM/A* and *VDIM/A* and *VIFM/A* are split into *VIAM/A* and *VIIM/A*.

### *Private Household Demand (8:6, 9:6)*

Taxes on purchases of commodities by the private households differ according to whether the commodities are domestically produced or imported, i.e.,

```
Variable (all,i,TRAD_COMM)(all,r,REG)
    tpd(i,r) # shock to tax on domestic i purchased by private hhld in r #;
Variable (all,i,TRAD_COMM)(all,r,REG)
    tpm(i,r) # shock to tax on imported i purchased by private hhld in r #;
```

hence there are separate tax revenues for domestic and imported commodities, i.e.,

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    DPTAX(i,r) # tax on private consumption of domestic good i in region r #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    DPTAX(i,r) = VDPA(i,r) - VDPM(i,r);
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    IPTAX(i,r) # tax on private consumption of imported good i in region r #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    IPTAX(i,r) = VIPA(i,r) - VIPM(i,r);
Coefficient (all,r,REG)
    TPC(r) # private consumption tax payments in r #;
Formula (all,r,REG)
    TPC(r) = sum(i,TRAD_COMM, DPTAX(i,r) + IPTAX(i,r));
```

### But taxes are defined by

```
Equation PHHDPRICE
# eq'n links domestic market and private consumption prices (HT 18) #
(all,i,TRAD_COMM)(all,r,REG)
    ppd(i,r) = atpd(i,r) + pm(i,r);
Equation PHHIPRICES
# eq'n links domestic market and private consumption prices (HT 21) #
(all,i,TRAD_COMM)(all,r,REG)
    ppm(i,r) = atpm(i,r) + pim(i,r);
```

where

```
Variable (all,i,TRAD_COMM)(all,r,REG)
    atpm(i,r) # actual tax on imported i purchased by private hhlds in r #;
Variable (all,r,REG)
    tp(r) # region-wide shock to tax on purchases by private hhld in r #;
Equation TPDSHIFT
# eq'n permits implementation of a uniform consumption tax change #
(all,i,TRAD_COMM)(all,r,REG)
    atpd(i,r) = tpd(i,r) + tp(r);
```

where *ppd* and *ppm* define the prices paid by the private households while *pm* and *pim* define the price received by the suppliers. Hence *VDPM* and *VIPM* record the transactions between the commodity and private household accounts and *DPTAX* and *IPTAX* define the associated tax payments by private households to the government.

### *Government Demand (8:13, 9:13)*

Taxes on purchases of commodities by the government differ according to whether the commodities are domestically produced or imported, i.e.,

```
Variable (all,i,TRAD_COMM)(all,r,REG)
    tgd(i,r) # tax on domestic i purchased by government hhlds in r #;
Variable (all,i,TRAD_COMM)(all,r,REG)
    tgm(i,r) # tax on imported i purchased by gov't hhld in r #;
```

hence there are separate tax revenues for domestic (DGTAX) and imported (IGTAX) commodities, i.e.,

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    DGTAX(i,r) # tax on government consumption of domestic good i in region r #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    DGTAX(i,r) = VDGA(i,r) - VDGM(i,r);
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    IGTAX(i,r) # tax on government consumption of imported good i in region r #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    IGTAX(i,r) = VIGA(i,r) - VIGM(i,r);
Coefficient (all,r,REG)
    TGC(r) # government consumption tax payments in r #;
Formula (all,r,REG)
    TGC(r) = sum(i,TRAD_COMM, DGTAX(i,r) + IGTAX(i,r));
```

But taxes are defined by

```
Equation GHHDPRICE
# eq'n links domestic market and government consumption prices (HT 19) #
(all,i,TRAD_COMM)(all,r,REG)
    pgd(i,r) = tgd(i,r) + pm(i,r);
Equation GHHIPRICES
# eq'n links domestic market and government consumption prices (HT 22) #
(all,i,TRAD_COMM)(all,r,REG)
    pgm(i,r) = tgm(i,r) + pim(i,r);
```

where *pgd* and *pgm* define the prices paid by the government while *pm* and *pim* define the price received by the suppliers. Hence *VDGM* and *VIGM* record the transactions between the commodity and private household accounts and *DGTAX* and *IGTAX* define the associated tax payments by government to the government.

### Factor Use Taxes (10:3)

These are taxes on factor use paid by the activities. The GTAP tablo file defines these as follows

```
Coefficient (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
    ETAX(i,j,r) # tax on use of endowment good i by industry j in region r #;
Formula (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
    ETAX(i,j,r) = VFA(i,j,r) - VFM(i,j,r);
Coefficient (all,r,REG)
    TFU(r) # firms' tax payments on primary factor usage in r #;
Formula (all,r,REG)
    TFU(r) = sum(i,ENDW_COMM, sum(j,PROD_COMM, ETAX(i,j,r)));
```

where

```
Coefficient (all,i,DEMD_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) # producer expenditure on i by j in r valued at agents' prices #;
Formula (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) = EVFA(i,j,r);
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
    VFA(i,j,s) = VDFA(i,j,s) + VIFA(i,j,s);
```

The coefficient *ETAX* is defined over three sets

- i) ENDW\_COMM – the factors of production (*f* in GAMS);
- ii) PROD\_COMM – the activities (*a* in GAMS) and investment goods (“*kap*” in GAMS);
- iii) REG – the regions (*k* in GAMS).

However note that while *VFM* is defined over the same three sets, *VFA* is defined over three sets, one of which is different, i.e.,

- i) DEMD\_COMM – the factors of production (*f* in GAMS) plus the activities (*a* in GAMS), i.e., ENDW\_COMM is a subset of DEMD\_COMM;
- ii) PROD\_COMM – the activities (*a* in GAMS) and investment goods (“*kap*” in GAMS);
- iii) REG – the regions (*k* in GAMS).

Hence only part of *VFA* is used in the determination of the factor use taxes, and in the GTAP model *VFA* is derived from the read in data. Furthermore, note that there are no entries for CGDS in either *VFM* or *EVFA*.

Hence we define  $EVFA2_{f,a,k}$  to be equal to  $EVFA_{endw\_comm,trad\_comm,reg}$ , i.e., exclusive of the column entries for ‘CGDS’. Similarly we define  $VFM2_{f,a,k}$  to be equal to  $VFMA_{endw\_comm,trad\_comm,reg}$ , i.e., exclusive of the column entries for ‘CGDS’. Then taxes paid by activities for the use of factors are defined as

$$FACTTAX_{f,a,k} = EVFA2_{f,a,k} - VFM2_{f,a,k}.$$

The price relationships in the GTAP model define

```
Equation MPFACTPRICE
# eq'n links domestic and firm demand prices (HT 16) #
(all,i,ENDWM_COMM)(all,j,PROD_COMM)(all,r,REG)
  pfe(i,j,r) = tf(i,j,r) + pm(i,r);
```

where *pfe* is the price paid by the firm and *pm* the price received by the factor. Hence *VFM2* defines payments by activities to factors while *FACTTAX* defines payments to the government, as factor use taxes, by activities.

‘Production’ Taxes (11:3, 12:4 and 11:14)

These are the taxes identified by the parameter *PTAX* in the GTAP model. *PTAX* is defined over two sets

1. NSAV\_COMM – the factors of production (*f* in GAMS), the activities (*a* in GAMS) and investment goods (“*kap*” in GAMS);

## 2. REG – the regions (*k* in GAMS);

and is assigned values from two parameters, *VOM* and *VOA*, that are defined over the two same sets and are created from various components of the read in data, i.e.,

```
Coefficient (all,i,NSAV_COMM)(all,r,REG)
    PTAX(i,r) # output tax on good i in region r #;
Formula (all,i,NSAV_COMM)(all,r,REG)
    PTAX(i,r) = VOM(i,r) - VOA(i,r);
Coefficient (all,r,REG)
    TOUT(r) # production tax payments in r #;
Formula (all,r,REG)
    TOUT(r) = sum(i,PROD_COMM, PTAX(i,r));
```

While the parameters/coefficients *VOM* and *VOA* are defined over 2 sets they are assigned values from a number of sources whose values are defined over different sets. Since these different sets ultimately refer to different sub matrices of the SAM it will prove convenient to decompose the coefficient *PTAX* and therefore the coefficients *VOA* and *VOM*. The set for *i* in *PTAX* suggests that *PTAX* will be paid by three types of agents in the SAM – activities, factors and the investment account – hence the SAM should allow for non-zero entries in the sub matrices 11:3, 12:4 and 11:14.<sup>10</sup>

*VOA* is relatively straightforward since it only uses 2 subsets of *NSAV\_COMM*.

```
Coefficient (all,i,NSAV_COMM)(all,r,REG)
    VOA(i,r) # value of commodity i output in region r at agents' prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
    VOA(i,r) = EVOA(i,r);
Formula (all,i,PROD_COMM)(all,r,REG)
    VOA(i,r) = sum(j,DEMD_COMM, VFA(j,i,r));
```

with

```
Coefficient (all,i,DEMD_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) # producer expenditure on i by j in r valued at agents' prices #;
Formula (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) = EVFA(i,j,r);
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
    VFA(i,j,s) = VDFFA(i,j,s) + VIFFA(i,j,s);
```

*VOM* draws on more subsets of *NSAV\_COMM*. i.e.,

```
Coefficient (all,i,NSAV_COMM)(all,r,REG)
    VOM(i,r) # value of commodity i output in region r at market prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
    VOM(i,r) = sum(j,PROD_COMM, VFM(i,j,r));
Formula (all,m,MARG_COMM)(all,r,REG)
    VOM(m,r) = VDM(m,r) + sum(s,REG, VXMD(m,r,s)) + VST(m,r);
Formula (all,i,NMRG_COMM)(all,r,REG)
    VOM(i,r) = VDM(i,r) + sum(s,REG, VXMD(i,r,s));
Formula (all,h,CGDS_COMM)(all,r,REG)
    VOM(h,r) = VOA(h,r);
```

with

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) # domestic sales of i in r at mkt prices (tradeables only) #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) = VDPM(i,r) + VDGM(i,r) + sum(j,PROD_COMM, VDFM(i,j,r));
```

<sup>10</sup> Note that the references to cells in Table 4 treat two of these taxes as indirect taxes and the third as a direct tax. This is explained in the discussion of *PTAXFACT*.

The general equation for the *PTAX* is

```
Equation OUTPUTPRICES
# eq'n links pre- and post-tax supply prices for all industries (HT 15) #
(all,i,PROD_COMM)(all,r,REG)
  ps(i,r) = to(i,r) + pm(i,r);
!<
  This equation links pre- and post-tax supply prices for all industries.
  This captures the effect of output taxes. TO(i,r) < 1 in the case of a
  tax. (HT 15)
>!
```

where

```
Variable (all,i,NSAV_COMM)(all,r,REG)
  to(i,r) # output (or income) tax in region r #;
```

Hence all transactions that go into *PTAX* should be valued in terms of *ps* and *pm*.

It is convenient to deal with *PTAX* in three parts; one each for activities, investment goods and factors. The endowment/factor component is most straightforward.

#### *Factor Income Taxes* (12:4)

*PTAX* for factors – *PTAXFACT* – is defined over the sets for factors (*f*) and regions (*r*) as

```
Formula (all,i,NSAV_COMM)(all,r,REG)
  PTAX(i,r) = VOM(i,r) - VOA(i,r);
Formula (all,i,ENDW_COMM)(all,r,REG)
  VOA(i,r) = EVOA(i,r);
Formula (all,i,ENDW_COMM)(all,r,REG)
  VOM(i,r) = sum(j,PROD_COMM, VFM(i,j,r));
```

where

```
Coefficient (ge 0)(all,i,ENDW_COMM)(all,r,REG)
  EVOA(i,r) # value of commodity i output in region r #;
Update (all,i,ENDW_COMM)(all,r,REG)
  EVOA(i,r) = ps(i,r) * qo(i,r);
Read
  EVOA from file GTAPDATA header "EVOA";
```

and

```
Coefficient (ge 0)(all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) # producer expenditure on i by j in r valued at mkt prices #;
Update (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) = pm(i,r) * qfe(i,j,r);
Update (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
  VFM(i,j,r) = pmes(i,j,r) * qfe(i,j,r);
Read
  VFM from file GTAPDATA header "VFM";
```

```
Equation FACTORINCPRICES
# eq'n links pre- and post-tax endowment supply prices (HT 15) #
(all,i,ENDW_COMM)(all,r,REG)
  ps(i,r) = to(i,r) + pm(i,r);
```

Hence the factor income tax transactions can be defined as

$$\begin{aligned}
 PTAXFACT_{f,k} &= \sum_{prod\_comm} VFM_{endw\_comm,prod\_comm,reg} - EVOA_{endw\_comm,reg} \\
 &= \sum_a VFM2_{f,a,k} - EVOA_{f,k}
 \end{aligned}$$

where  $VFM2$  is defined as  $VFM$  excluding the entries for 'CGDS'. Hence these are taxes paid by factors direct to the government, and, in terms of the SAM relationships, are defined as the difference between payments to factors and the disposable income of factors after payment of depreciation and taxes, i.e., factor income taxes.

### *Taxes on Production (11:3)*

$PTAX$  for activities –  $PRODTAX$  – is defined over the sets for activities ( $a$ ) and regions ( $k$ ) as

```
Formula (all,i,NSAV_COMM)(all,r,REG)
    PTAX(i,r) = VOM(i,r) - VOA(i,r);
Formula (all,i,PROD_COMM)(all,r,REG)
    VOA(i,r) = sum(j,DEMD_COMM, VFA(j,i,r));
Formula (all,m,MARG_COMM)(all,r,REG)
    VOM(m,r) = VDM(m,r) + sum(s,REG, VXMD(m,r,s)) + VST(m,r);
Formula (all,i,NMRG_COMM)(all,r,REG)
    VOM(i,r) = VDM(i,r) + sum(s,REG, VXMD(i,r,s));
```

with

```
Coefficient (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) # domestic sales of i in r at mkt prices (tradeables only) #;
Formula (all,i,TRAD_COMM)(all,r,REG)
    VDM(i,r) = VDPM(i,r) + VDGM(i,r) + sum(j,PROD_COMM, VDFM(i,j,r));
Coefficient (all,i,DEMD_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) # producer expenditure on i by j in r valued at agents' prices #;
Formula (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)
    VFA(i,j,r) = EVFA(i,j,r);
Formula (all,i,TRAD_COMM)(all,j,PROD_COMM)(all,s,REG)
    VFA(i,j,s) = VDFFA(i,j,s) + VIFA(i,j,s);
```

$VOM$  has already been defined over the  $TRAD\_COMM$  set, but it remains to define  $VOA$  over this set. Working with  $PROD\_COMM$  for the moment, the formula above indicates that  $VOA$  is derived from  $VFA$ , where  $VFA$  defines matrices of dimension  $DEMD\_COMM$  and  $PROD\_COMM$  for each region  $r$ . The component of  $VFA$  referring to  $ENDW\_COMM$  comes from  $EVFA$  while the components referring to  $PROD\_COMM$  come from  $VDFFA$  and  $VIFA$ . Hence

$$VOA_{a,k} = \sum_f EVFA_{f,a,k} + \sum_{dc} VDFFA_{dc,a,k} + \sum_{mc} VIFA_{mc,a,k}$$

which simply defines the value of output by activity  $a$  in region  $r$  as the sum of the payments to primary inputs at agents prices ( $EVFA$ ) plus the sum of domestic ( $VDFFA$ ) and imported ( $VIFA$ ) inputs valued at agents prices. But  $VOM$  is defined as

$$VOM_{dc,k} = VDM_{dc,k} + VXMD_{dc,k} + VST_{m,k}$$

which suggests that the dimension of  $VOA$  and  $VOM$  are non commensurate. However the elements of  $VOM$  are the entries on the diagonal supply matrix and hence,

$$VOM_{dc,k} = VOM_{a,k} \cdot$$

### *Taxes on Investment (11:14)*



*PATX* for investment – *PTAXINV* – is straightforward. The expression for the power of the tax comes with a warning

```
Variable (all,i,NSAV_COMM)(all,r,REG)
  to(i,r) # output (or income) tax in region r #;
!<
  Note: it is important that the user NOT shock the tax on capital goods
  output, as this will cause an inconsistency in the update relationship
  for VOM(cgds).
>!
```

This indicates that there is no facility for *PTAXINV* in the GTAP model. A conclusion that is confirmed by the formula for the output valuations at market and agents prices for capital goods, i.e.,

```
Formula (all,h,CGDS_COMM)(all,r,REG)
  VOM(h,r) = VOA(h,r);
```

Hence by definition *PTAXINV* is zero.

### *Taxes and Regional Household Income (5:7, 5:8, 5:8, 5:10, 5:11, 5:12)*

All the tax instruments discussed above are expressed in terms of the expenditures on taxes by different agents in different contexts, and in this SAM all these taxes are paid to accounts relating to particular tax instruments. To convert these into incomes to the regional household it is only necessary to compute the row total for each tax instrument and transpose these as row entries into the regional household account.

### Trade Accounts

Trade transactions are at the heart of the GTAP database. A distinctive feature of version 5 of the GTAP database is the recording of trade margins; in essence this allows the identification of origin specific trade and transport costs associated with imports of commodities by each region with each region contributing to a global trade and transport ‘pool’ from which trade and transport services are drawn. Consequently while the use of trade and transport services associates the use of services with specific imports the specific suppliers of these services are not identified.

### *Imports (15:1, 16:5)*

Imports at world prices (*VIWS*) are valued carriage, insurance and freight (cif) paid, i.e.,

```
Variable (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  pcif(i,r,s) # CIF world price of commodity i supplied from r to s #;
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VIWS(i,r,s) # imports of i from r to s valued cif (tradeables only) #;
Update (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VIWS(i,r,s) = pcif(i,r,s) * qxs(i,r,s);
Read
  VIWS from file GTAPDATA header "VIWS";
```

where

```
Variable (orig_level=VXMD)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    qxs(i,r,s) # export sales of commodity i from r to region s #;
```

while imports at market prices (*VIMS*) are valued inclusive of import duties (see the discussion of import duties in the section on indirect taxes)

```
Variable (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    pms(i,r,s) # domestic price for good i supplied from r to region s #;
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    VIMS(i,r,s) # imports of i from r to s valued at domestic mkt prices #;
Update (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    VIMS(i,r,s) = pms(i,r,s) * qxs(i,r,s);
Read
    VIMS from file GTAPDATA header "VIMS";
```

Hence *VIWS* defines the import transactions inclusive of trade and transport costs. The

GTAP model defines

```
Coefficient (ge 0) (all,m,MARG_COMM)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    VTMFSD(m,i,r,s)
    # int'l margin usage, by margin, freight, source, and destination #;
Update (all,m,MARG_COMM)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    VTMFSD(m,i,r,s) = pt(m) * qtmfsd(m,i,r,s);
Read
    VTMFSD from file GTAPDATA header "VTWR";
```

where

```
Variable (all,m,MARG_COMM)
    pt(m) # price of composite margins services, type #;
Variable (all,m,MARG_COMM)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
    qtmfsd(m,i,r,s) # international usage margin m on i from r to s #;
```

such that *VTWR* records the transactions on margin services (*m*) associated with commodity *i* exported by region *r* to region *s*.

Hence transactions with other regions for imported commodities are defined as (*VIWS* – *VTWR*) and transactions with the global pool of transport services associated with these imports are defined by *VTWR*.

However, this direct method of calculating the value of import transactions (*fob*) does encounter a problem for a programme in GAMS. Specifically imports (*fob*) of commodity *c* from region *r* by region *s* are not identical to the exports (*fob*) of commodity *c* to region *r* from region *s*. The differences are minimal – most only appear at the fourth or more decimal places – but they do create some problems when verifying that the global SAM is complete and consistent and they impede subsequent transformations of the global SAM. To avoid these potential problems the values of import transaction (*fob*) are actually calculated indirectly; specifically imports (*fob*) are computed by transposing the 3-dimensional matrix of exports (*fob*) (for details see below). This ensures that the values of imports and exports are identical, which avoids any subsequent problems.

### *Exports (2:16, 2:17)*

Export transactions are recorded at free on board (fob) prices, i.e., at the price paid by the importing region before the payment of transports costs. The GTAP databases records this as

#### *VXWD*

```
Variable (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  pfob(i,r,s) # FOB world price of commodity i supplied from r to s #;
Coefficient (ge 0)(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXWD(i,r,s) # exports of i from r to s valued fob (tradeables only) #;
Update (all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  VXWD(i,r,s) = pfob(i,r,s) * qxs(i,r,s);
Read
  VXWD from file GTAPDATA header "VXWD";
```

where

```
Equation EXPRICES
# eq'n links agent's and world prices (HT 27) #
(all,i,TRAD_COMM)(all,r,REG)(all,s,REG)
  pfob(i,r,s) = pm(i,r) - tx(i,r) - txs(i,r,s);
```

and hence the world price (*pfob*) differs from the market price (*pm*) by the export taxes (*txs*).<sup>11</sup>

But the database records separately the exports of transport services as

```
Variable (orig_level=VST)(all,m,MARG_COMM)(all,r,REG)
  qst(m,r) # sales of m from r to international transport #;
Coefficient (ge 0)(all,m,MARG_COMM)(all,r,REG)
  VST(m,r)
  # expts of m from r for int'l trnsport valued at mkt p (tradeables only) #;
Update (all,m,MARG_COMM)(all,r,REG)
  VST(m,r) = pm(m,r) * qst(m,r);
Read
  VST from file GTAPDATA header "VST";
```

note that the price (*pm*) on exports of transport services is a market price, i.e., there is no allowance for export taxes on transport services.

### *Margins*

The transport margins have been identified above but it also worth noting that the relationships between imports and exports also require an explicit balancing of the fob exports from a source region with the cif imports and margins by a destination region, i.e.,

```
In a balanced data base, (all,i,TRAD_COMM)(all,r,REG)(all,s,REG),
  VIWS(i,r,s) = VXWD(i,r,s) + VTFSD(i,r,s)
```

### Capital Account (14:4, 14:5, 14:16, 14:17)

Income to the capital account consists of four elements, two of which are related to the domestic accounts and two to the trade accounts. Domestic savings consist of the allowance for depreciation (see above) and net domestic saving, which is defined in the GTAP tablo file as

```

Variable (all, r, REG)
    psave(r) # price of savings in region r #;
Variable (all,r,REG)
    qsave(r) # regional demand for NET savings #;
Coefficient (all,r,REG)
    SAVE(r) # expenditure on NET savings in region r valued at agent's prices #;
Update (all,r,REG)
    SAVE(r) = psave(r) * qsave(r);
Read
    SAVE from file GTAPDATA header "SAVE";

```

Following the convention in the GTAP model all savings are assumed to originate from the regional household.

This leaves the balance on the capital account. Strictly this could be defined as the differences in the values of exports fob and import cif, but the GTAP tablo file treats trade margins through the device of a global transport pool, i.e., while specific transport transactions are associated with imports from specific origins, they are not mapped onto the region that supplies the transport services. Rather transport services are exported to a global transport pool. Hence the balance on the capital account with the rest of the world consists of elements relating to both commodities and transport services. The capital account balance with each trading partner is defined as the value of imports valued cif less the value of transport services less the value of export. The capital account balance with the global transport pool is defined as the values of imported transport services less the value of exported transport services.

## 5 Balancing the Global SAM

It is essential to verify that the global SAM is ‘exactly’ balanced. This is carried out in Section 9 of the files `gtapsam*.gms` and involves two related sets of checks. First, it is necessary to check that the row and column totals of each regional SAM are identical, and second it is necessary to verify that each and every component of the transactions between regions is identical to its matching counterpart. If any differences are identified it is necessary to determine whether these are a consequence of ‘errors’ in the base data or a result of ‘errors’ introduced when developing the global SAM.

The resultant global SAM is not precisely balanced, i.e., the row and column totals differ, although the extent of the imbalance is very small.<sup>12</sup> Moreover **all** the imbalances are related to the imported commodity accounts and the Regional Household and Capital (savings and investment) accounts.

---

<sup>11</sup> *tx* is included for simulation purposes only and hence is redundant in the current context.

**Table 5a Absolute and Percentage Differences in Row and Column Totals in the SAM - Version 5.0 of the GTAP Database**

Absolute Difference		Percentage Difference		Absolute and Percent Difference		
Critical Value	Cases	Critical Value	Cases	Critical Values		Cases
				<i>Absolute</i>	<i>Percent</i>	
5	0	5	19			
1	0	1	29			
0.5	2	0.5	32	1	0.1	0
0.1	8	0.1	49	0.1	0.01	0
0.01	75	0.01	89	0.01	0.001	0
0.001	295	0.001	167	0.001	0.0001	3
0.0001	1,173	0.0001	333	0.0001	0.00001	845
0.00001	2,309	0.00001	2,303	0.00001	0.000001	2,258
0.000001	3,360	0.000001	3,515	0.000001	0.0000001	3,329

Source: Authors calculations.

Tables 4a and 4b provide some summary statistics about the absolute and proportionate differences between the row and column totals for the SAMs derived from versions 5.0 and 5.3 of the GTAP database. For Version 5.0 the SAM has 46,200 accounts, i.e., 66 regions with 700 accounts per region, for which differences in row and column totals were recorded for only 3,664 accounts. While for Version 5.3 the SAM has 60,216 accounts, i.e., 78 regions with 772 accounts per region, for which differences in row and column totals were recorded for only 4,306 accounts. Both Tables confirm that while there are an apparently appreciable number of cases for which there are absolute and percentage differences in the accounts totals for a relative small proportion of the SAM account, there are a negligible number of cases for which the absolute differences are greater than 0.001 **and** the percentage differences are greater than 0.0001 percent.

<sup>12</sup> The differences between the row and column totals are very small and consequently are unlikely to be relevant when solving a model, especially when solved to single precision as in GEMPACK.

**Table 5b**                      **Absolute and Percentage Differences in Row and Column Totals in the SAM - Version 5.3 of the GTAP Database**

Absolute Difference		Percentage Difference		Absolute and Percent Difference		
Critical Value	Cases	Critical Value	Cases	Critical Values		Cases
				<i>Absolute</i>	<i>Percent</i>	
5	0	5	15			
1	0	1	28			
0.5	2	0.5	33	1	0.1	0
0.1	11	0.1	54	0.1	0.01	0
0.01	70	0.01	99	0.01	0.001	0
0.001	315	0.001	211	0.001	0.0001	5
0.0001	1,192	0.0001	456	0.0001	0.00001	883
0.00001	2,540	0.00001	2,753	0.00001	0.000001	2,481
0.000001	3,907	0.000001	4,138	0.000001	0.0000001	3,877

Source: Authors calculations.

Moreover, since all the checks on the data indicate that the differences in the row and column totals can be traced back to the GTAP database, it can be safely concluded that they are not consequences of errors in the conversion to a SAM representation, rather that they are differences that are within the tolerances set by the compilers of the database. But, even minor differences in total incomes and expenditures can cause modelling difficulties; hence routines are included in the programmes to resolve these differences.<sup>13</sup>

#### Balancing the SAMs from Versions 5.0 and 5.3 of the GTAP Database

There are four distinct categories of ‘errors’ in the database. These were resolved sequentially with the differences in the row and column totals for each regions being recalculated after each group of ‘errors’ was corrected (see Section 10 of the file *gtapdat6.gms*) and then the global SAM was rechecked to ensure the final SAM was exactly balanced (see Section 11 of the files *gtpsam\*.gms*).

The ‘errors’ and the means of adjusting them are detailed below.

- i) No private household demands for imports of dwellings (account *mdwe*) were recorded in some regions. The recorded transactions for the other regions (0.000001) appear to have been included to ensure that some transactions were

<sup>13</sup> If these routines are not wanted then the user can switch off the relevant sections of the programme, i.e., sections 10 and 11.

recorded. The *mdwe* transactions in all regions were therefore set equal to 0.000001.

- ii) The trade accounts were left unchanged and the differences resolved by changing Government final demand transactions so as to remove the differences.
- iii) The row and column totals for the regional household accounts differ and these differences are matched by equal and opposite differences in the row and column totals for the capital accounts. Again these were minor differences, and they were corrected by adjusting the transfers from the regional households to the private households, government and capital accounts.

## **6 Transactions Values in the Global SAMs**

The maximum values for transactions within each regional SAM for each version of the Database are reported in Tables 6a and 6b. For each and every region in both versions the minimum absolute value of transactions reported was 0.000001. This indicates that the proportionate differences between maximum and minimum transactions are very substantial, which suggests that for use in GAMS based models there is a *prime facie* case for reducing the ‘precision’ with which entries in the global SAM are recorded.

**Table 6a**                      **Maximum absolute value of transactions in SAM by region - Version 5.0 of the GTAP Database**

<b>Region</b>	<b>Max TV</b>	<b>Region</b>	<b>Max TV</b>	<b>Region</b>	<b>Max TV</b>
arg	254,987.300000	ind	258,450.000000	twm	172,051.700000
aus	240,439.000000	irl	37,969.670000	tza	5,919.954000
aut	115,852.500000	ita	718,484.900000	uga	5,507.293000
bel	153,784.300000	jpn	2,532,620.000000	ury	14,747.600000
bgd	35,002.410000	kor	237,332.900000	usa	5,495,106.000000
bra	508,587.200000	lka	10,912.530000	ven	58,484.790000
bwa	2,332.448000	lux	6,853.894000	vnm	15,468.870000
can	370,405.800000	mar	23,011.550000	xap	19,256.660000
che	149,266.700000	mex	262,957.200000	xce	84,781.290000
chl	50,492.250000	moz	3,217.104000	xcm	79,049.600000
chn	414,093.400000	mwi	2,148.372000	xef	77,843.990000
col	64,847.740000	mys	44,083.010000	xme	274,229.700000
deu	1,205,839.000000	nld	215,073.600000	xnf	113,888.300000
dnk	86,348.520000	nzl	41,337.540000	xrw	213,016.500000
esp	329,737.000000	per	43,070.480000	xsa	53,097.770000
fin	63,420.060000	phl	59,729.110000	xsc	82,846.240000
fra	834,165.600000	pol	87,076.330000	xsf	4,533.209000
gbr	847,541.100000	prt	65,164.870000	xsm	7,989.729000
grc	87,056.410000	sgp	60,809.450000	xss	124,257.100000
hkg	103,586.300000	swe	120,956.800000	xsu	363,176.600000
hun	28,459.320000	tha	91,607.060000	zmb	3,122.862000
idn	133,054.000000	tur	129,187.700000	zwe	5,272.513000

Source: Authors calculations from GTAP database.



**Table 6b**                      **Maximum absolute value of transactions in SAM by region – Version 5.3 of the GTAP Database**

<b>Region</b>	<b>Max TV</b>	<b>Region</b>	<b>Max TV</b>	<b>Region</b>	<b>Max TV</b>
col	64,847.75	bra	508,587.14	cze	28,090.54
aus	240,439.00	chl	50,492.25	rom	26,434.67
nzl	41,337.55	ury	14,747.60	svk	12,072.92
chn	414,093.43	xsm	7,989.73	svn	11,048.58
hkg	103,586.36	aut	115,852.48	hrv	11,774.25
jpn	2,532,619.73	dnk	86,348.50	alb	2,239.78
kor	237,332.90	fin	63,420.05	est	2,849.00
twn	172,051.71	fra	834,165.63	lva	3,100.14
idn	133,053.78	deu	1,205,838.57	ltu	5,717.13
mys	46,833.33	gbr	847,540.98	rus	271,856.11
phl	59,728.72	grc	87,056.38	xsu	81,099.67
sgp	67,347.21	irl	37,969.67	tur	129,187.74
tha	84,475.22	ita	718,484.77	xme	274,229.80
vnm	15,468.88	nld	215,073.41	mar	23,011.56
bgd	35,002.41	prt	65,164.02	xnf	113,888.28
ind	258,450.02	esp	329,737.16	bwa	2,321.43
lka	10,912.53	swe	120,956.83	xsc	82,846.23
xsa	53,097.76	bel	153,784.29	mwi	2,148.37
can	370,405.86	lux	6,853.89	moz	3,217.10
usa	5,495,105.65	che	149,266.71	tza	5,919.95
mex	262,957.19	xef	77,844.00	zmb	3,122.86
xcm	79,049.60	cyp	5,037.34	zwe	5,272.51
per	43,070.49	mlt	2,121.08	xsf	4,528.45
ven	58,484.79	hun	28,459.32	uga	5,507.29
xap	19,256.66	pol	87,076.35	xss	124,257.10
arg	254,987.29	bgr	9,545.76	xrw	192,746.37

Source: Authors calculations from GTAP database.

## Concluding Comments

This exercise has demonstrated that the GTAP database can be readily transformed into a global SAM and that, barring a few very small adjustments that have no modeling implications, the resultant SAM conforms to a standard definition of a SAM. Such a result is hardly surprising; since the data for all whole economy models can always be represented in a SAM format provided the models are complete and consistent, which is the case for the GTAP model. Nevertheless the exercise has several substantial benefits. First, it provides SAM databases that can be used by SAM based CGE modelers whose preferred programming language is GAMS. Second, it provides a method by which individual region SAM can be easily extracted from the GTAP database, which may make it easier for some researchers to

evaluate the data for specific regions. And third, it provides an alternative representation of the GTAP database that might be more easily interpreted by some users of the database.

There are clearly a number of subsequent developments that might prove useful. These, in no order of priority, might include:

- i) developing a Header Array (HAR) file version that can be used by GEMPACK users;
- ii) augmenting the database with additional transactions data, e.g., remittances (private and ‘official’) between regions; and
- iii) defining a series of reduced form representations of the SAM, e.g., versions that
  - a) remove the Regional Household Account; and
  - b) impose a ‘law of one price’ so that commodity income transactions are recorded at purchaser prices;
- iv) developing an aggregation programme in GAMS.

Several of these developments have been undertaken. The GAMS2HAR utility means that this global SAM can be written out as a HAR file, although the resulting HAR file is large (180 MB). Programmes that produce 6 different reduced forms of the database and carry out aggregations (developments iii and iv) are reported in McDonald (2003). One method by which the information in the GTAP database can be augmented, and an evaluation of the potential benefits of such data augmentation, has been developed, while a GEMPACK programme for converting the GTAP database to a SAM might be useful.

## References

- GTAP (2002). 'GTAP61.TAB',  
<http://www.gtap.agecon.purdue.edu/products/models/current.asp>
- Hertel, T.W., (1997). *Global Trade Analysis: Modeling and Applications*. Cambridge: Cambridge University Press.
- King, B.B., (1985). 'What is a SAM?', in Pyatt, G. and Round, J.I. (ed), *Social Accounting Matrices: A Basis for Planning*. Washington: World Bank.
- McDonald, S., (2004). 'Developing and Aggregating Reduced Form Social Accounting Matrices from the GTAP Database', *mimeo*.
- McDonald, S., and Sonmez, Y., (2004). 'Extending a Global GTAP Social Accounting Matrix to incorporate Inter-institutional and Inter-Regional Transactions', *mimeo*.
- Pyatt, G. and Round, J.I., (1977). 'Social Accounting Matrices for Development Planning', *Review of Income and Wealth*, Vol 23, pp 339-364.
- Pyatt, G., (1991). 'Fundamentals of Social Accounting', *Economic Systems Research*, Vol 3, pp 315-341.
- Reinert, K.A. and Roland-Holst, D.W., (1977). 'Social Accounting Matrices' in Francois, J.F. and Reinert, K.A., (eds) *Applied Methods for Trade Policy Analysis: A Handbook*. Cambridge University Press: Cambridge
- Rutherford, T.F., (1998). 'GTAPinGAMS: The Dataset and Static Model', *mimeo*.
- Sadoulet, E. and de Janvry, A., (1995). *Quantitative Development Policy Analysis*. John Hopkins University Press: Baltimore. Chap 10.
- UN, (1993). *System of National Accounts 1993*.

## Appendix 1 Programme Files

The GAMS code for transforming the HAR file representation of the GTAP database into a SAM representation saved in GDX format is provided as a GAMS model library. There are currently 3 versions of the code; the first works with version 5.0 of the GTAP database while the second and third with version 5.3. The difference between the 2 version that work with version 5.3 of the database is in the way the data are made readable by GAMS; the second version uses SEEHAR to transform the data while the third uses HAR2GAMS to make the data GAMS readable and GAMS2HAR to export a version of the global SAM to a HAR file. Each component of the library links to 6 or 7 files (a separate GTAP set file is no needed when using HAR2GAMS):

<b>GTAP Database 5.0</b> (using SEEHAR)	<b>GTAP Database 5.3</b> (using SEEHAR)	<b>GTAP Database 5.3</b> (using HAR2GAMS)
gtpsam50_2.gms gtapsamset50.inc gtapmap50.inc set57_66.inc gtapid57.gdx gpl.txt gtapsam.pdf	gtpsam53_2.gms gtapsamset53.inc gtapmap53.inc set57_78.inc gtapid57.gdx gpl.txt gtapsam.pdf	gtpsam53_3.gms gtapsamset53.inc gtapmap53_3.inc gtapid57.gdx gpl.txt gtapsam.pdf

All the files are in one WinZip archive – GTAPSAM.zip. Unzip this archive to a directory, say GTAPSAMLIB, and then access the required components using the model library gtaplib.glb that was included with in the archive.

To use the programme files follow the standard GAMSIDE procedures for a ‘User Model Library’. Note that it is also necessary to have a licensed version of the GTAP database with a current version of GTAPAgg (to extract the data and relabel the sector ‘for’ as ‘frs’) and a copy of SEEHAR (<http://www.monash.edu.au/policy/gpf90.htm>) and/or a copy of HAR2GAMS/GAMS2HAR (<http://debreu.colorado.edu/gdxhar/index.html>).

The procedures needed to use the programme are extensively documented at the start of the gms files. Please note that the authors are NOT in a position to provide support for users of these programmes.

## Warning

Be warned that the files created when using these programmes are large. For version 5.3 of the database the data file from SEEHAR is about 129MB, while the GDX file containing the SAM is about 19MB and the HAR file for a global SAM is 180MB. Depending on the DISPLAY options chosen the \*.lst file can exceed several hundred MB.

Furthermore the programme is not fast – it takes about 1 hour on a 200+MHz P4 with 512MB RAM. It would be possible to raise the speed by removing components that are not absolutely essential, but even so the resultant SAM will still contain 46,607,184 cells with approximately 3.1 million non zero entries.

## Appendix 2 Extract from the

### Programme

```
##### 8. CREATING THE GLOBAL SAM #####
*$ontext
* Constructing the global SAM
* Submatrices are added by rows of the SAM
* Number in brackets (x:y) indicate row (x) and column (y) in the SAM

##### Import commodity income components

* Import Intermediate Use matrices (1:3)

SAMG(mcc,aa,kk) = VIAM1(mcc,aa,kk) ;

* Household import demand vectors (1:6)

SAMG(mcc,"hhous",kk) = VIPM1(mcc,kk) ;

* Government import demand vectors (1:13)

SAMG(mcc,"ggovt",kk) = VIGM1(mcc,kk) ;

* Investment import demand vectors (1:14)

SAMG(mcc,"kkap",kk) = VIIM1(mcc,"kkap",kk) ;

##### Domestic commodity income components

* Domestic Intermediate Use matrices (2:3)

SAMG(dcc,aa,kk) = VDAM1(dcc,aa,kk) ;

* Household domestic demand vectors (2:6)

SAMG(dcc,"hhous",kk) = VDPM1(dcc,kk) ;

* Export taxes

SAMG(teetax,dcc,kk) = VXMD2(dcc,kk,teetax) -
VXWD2(dcc,kk,teetax) ;

* Government domestic demand vectors (2:13)

SAMG(dcc,"ggovt",kk) = VDGM1(dcc,kk) ;

* Investment domestic demand vectors (2:14)

SAMG(dcc,"kkap",kk) = VDIM1(dcc,"kkap",kk) ;
```

```
* Export demand matrices (2:16)
* Export of transport services

SAMG("ddotp","otppvst",kk) = VST1("ddotp",kk) ;
SAMG("ddwtp","wtppvst",kk) = VST1("ddwtp",kk) ;
SAMG("ddatp","atppvst",kk) = VST1("ddatp",kk) ;

* Exports of Goods and Services

SAMG(dcc,ww,kk) = VXWD1(dcc,kk,ww) ;

##### Activity Income

* Make Matrix (3:2)

SAMG(aa,dcc,kk) = (VDM(dcc,kk) + VST1(dcc,kk) +
SUM(ww,VXWD1(dcc,kk,ww))
- SUM(teetax,(VXMD2(dcc,kk,teetax) -
VXWD2(dcc,kk,teetax))))
* ID(aa,dcc) ;

##### Factor Income

* Factor demands by activities (4:3)

SAMG(ff,aa,kk) = VFM2(ff,aa,kk) ;

##### Indirect Tax Receipts for INDTAX(r) in GTAP model

* Import Duty Income (7:1)

SAMG(tmmtax,mcc,kk) = VIMS2(mcc,tmmtax,kk) -
VIWS2(mcc,tmmtax,kk) ;

*# Import Sales Taxes

* Intermediate Inputs (8:3)

SAMG(smmtax,aa,kk) = VIAA2(smmtax,aa,kk) - VIAM2(smmtax,aa,kk) ;

* Household Demand (9:6)

SAMG(smmtax,"hhous",kk) = VIPA2(smmtax,kk) - VIPM2(smmtax,kk) ;

* Government Demand (9:13)

SAMG(smmtax,"ggovt",kk) = VIGA2(smmtax,kk) - VIGM2(smmtax,kk) ;

* Investment Demand (9:14)
```



```

* Import Margin Matrices (15:1)
SAMG(OTPPMARG,mcc,kk) = VTWR2("otp",mcc,otppmarg,kk) ;
SAMG(WTPPMARG,mcc,kk) = VTWR3("wtp",mcc,wtpmarg,kk) ;
SAMG(ATPPMARG,mcc,kk) = VTWR4("atp",mcc,atppmarg,kk) ;

* Import Supply Matrices (16:1)

* SMcD TRADE

* Direct method

* SAMG(ww,mcc,kk) = VIWS1(mcc,ww,kk)
*
SUM(marg_comm,VTWR1(marg_comm,mcc,ww,kk)) ;

* Indirect Method using Exports fob

Parameter
SAMEExports1(mcc,ww,rr) Exports of commod mcc to region ww by
region rr
SAMEExports2(mcc,kk,rr) Exports of commod mcc to region kk by
region rr
SAMEExports3(mcc,kk,ww) Exports of commod mcc to region kk by
region ww

;

SAMEExports1(mcc,ww,rr) = SUM(dcc,kk)
$(MAPCOM(mcc,dcc)
$MAPREG(rr,kk)),
SAMG(dcc,ww,kk)) ;

SAMEExports2(mcc,kk,rr) = SUM(ww
$MAPREG4(kk,ww),
SAMEExports1(mcc,ww,rr)) ;

SAMEExports3(mcc,kk,ww) = SUM(rr
$MAPREG5(ww,rr),
SAMEExports2(mcc,kk,rr)) ;

SAMG(ww,mcc,kk) = SAMEExports3(mcc,kk,ww) ;

* Expenditures on transport services

SAMG("otppvst",otppmarg,kk) = SUM(mcc,SAMG(otppmarg,mcc,kk)) ;
SAMG("wtppvst",wtpmarg,kk) = SUM(mcc,SAMG(wtpmarg,mcc,kk)) ;
SAMG("atppvst",atppmarg,kk) = SUM(mcc,SAMG(atppmarg,mcc,kk)) ;

* Capital Account Balances
* Capital Account balance for trade in transport services

```

```

SAMG("kkap","otppvst",kk) =
SUM((mcc,otppmarg),SAMG(otppmarg,mcc,kk))
- SUM(dcc,SAMG(dcc,"otppvst",kk)) ;

SAMG("kkap","wtppvst",kk) =
SUM((mcc,wtpmarg),SAMG(wtpmarg,mcc,kk))
- SUM(dcc,SAMG(dcc,"wtppvst",kk)) ;

SAMG("kkap","atppvst",kk) =
SUM((mcc,atppmarg),SAMG(atppmarg,mcc,kk))
- SUM(dcc,SAMG(dcc,"atppvst",kk)) ;

* Capital Account balances by trade partner on commodity trade

SAMG("kkap",ww,kk) = SUM(mcc,SAMG(ww,mcc,kk))
- SUM(dcc,VXWD1(dcc,kk,ww)) ;

```